summon item_display ~ ~ ~ {Tags:[summon5,"UI-gamerule_control.ctr","UI-gamerule_control"]}
$execute as @n[tag=summon5] run function util:entity_owner/_this_set_a_owner {UUID:$(UUID)}
data modify entity @n[tag=summon5] teleport_duration set value 5

#SG :SelectedGamerule
$summon marker ~ ~ ~ {Tags:[summon6,"UI-gamerule_control.ctr.DM","UI-gamerule_control",util.noriding_kill],data:{SG:$(SG),id:$(id)}}
ride @n[tag=summon6] mount @n[tag=summon5]
$execute as @n[tag=summon6] run function util:entity_owner/_this_set_a_owner {UUID:$(UUID)}

summon interaction ~ ~ ~ {Tags:[summon7,"UI-gamerule_control.ctr.IA","UI-gamerule_control",util.noriding_kill],response:true}
ride @n[tag=summon7] mount @n[tag=summon5]
$execute as @n[tag=summon7] run function util:entity_owner/_this_set_a_owner {UUID:$(UUID)}

summon text_display ~ ~ ~ {Tags:[summon8,"UI-gamerule_control.ctr.TD","UI-gamerule_control",util.noriding_kill],billboard:"vertical"}
data modify entity @n[tag=summon8] transformation.translation[1] set value 1.05f
ride @n[tag=summon8] mount @n[tag=summon5]
$execute as @n[tag=summon8] run function util:entity_owner/_this_set_a_owner {UUID:$(UUID)}

$execute as @n[tag=summon5] run function dab:system/ui/gamerules/update_display {SG:$(SG)}


tag @n[tag=summon5] remove summon5
tag @n[tag=summon6] remove summon6
tag @n[tag=summon7] remove summon7
tag @n[tag=summon8] remove summon8